

# Adaptive Estimation of Emotion Generation for an Ambient Agent Model

Tibor Bosse, Zulfiqar A. Memon, and Jan Treur

Vrije Universiteit Amsterdam, Department of Artificial Intelligence,  
de Boelelaan 1081a, 1081 HV Amsterdam, The Netherlands  
{tbosse, zamemon, treur}@few.vu.nl  
<http://www.few.vu.nl/~{tbosse, zamemon, treur}>

**Abstract.** To improve the performance and wellbeing of humans in complex human-computer interaction settings, an interesting challenge for an ambient (or pervasive) agent system is to recognise the emotions of humans. To this end, this paper introduces a computational model to estimate the process of emotion generation based on certain triggers. The model has been implemented and tested using the modelling language LEADSTO. A first evaluation indicates that the model is successful in estimating a person's emotions, and is robust to different parameter settings.

## 1 Introduction

Ambient Intelligence [1], [2] represents a vision of the future where we will be surrounded by pervasive and unobtrusive electronic environments, which are sensitive, and responsive to humans. Such an environment has a certain degree of awareness of the presence and states of living creatures in it, and supports their activities. It analyses their behaviour, and may anticipate on it. Ambient Intelligence (AmI) integrates concepts from ubiquitous computing and Artificial Intelligence (AI) with the vision that technology will become invisible, embedded in our natural surroundings, present whenever we need it, attuned to the humans' senses, and adaptive to them. In an ambient intelligent environment, people are surrounded by networks of embedded intelligent devices that can sense their state, anticipate, and when relevant adapt to their needs. Therefore, the environment should be able to determine which actions have to be undertaken in order to keep this state optimal. For this purpose, it has to be equipped with knowledge about the relevant physiological and/or psychological aspects of human functioning.

In Cognitive Science and many other human-directed scientific areas (such as psychology, neurosciences, and biomedical sciences), models have been and are being developed for a variety of aspects of human functioning, among which visual attention, emotional processes, stress and workload. If such models of human processes are represented in a formal and computational format, and incorporated in the human environment in devices that monitor the physical and mental state of the human (cf. [19]) then such devices are able to perform a more in-depth analysis of the human's functioning. This can result in an environment that may more effectively affect

the state of humans by undertaking - in a knowledgeable manner - certain actions that improve their wellbeing and performance. For example, the workspaces of naval officers may include systems that, among others, track their eye movements and characteristics of incoming stimuli (e.g., airplanes on a radar screen), and use this information in a computational model that is able to estimate where their attention is focused at; see [8]. When it turns out that an officer neglects parts of a radar screen, such a system can either indicate this to the person or arrange on the background that another person or computer system takes care of this neglected part. In these types of applications, an ambience is created that has a better awareness and understanding of humans, based on computationally formalised knowledge from the human-directed disciplines.

Within the last decade, the literature in Cognitive Science and Artificial Intelligence shows an increasing amount of attempts to develop (computational) models of processes related to emotion [3]. In general, two classes of approaches can be distinguished: those that focus on *emotion elicitation* processes e.g., [5], [13], and those that focus on *emotion regulation* (or coping) processes e.g., [10], [20]. The first process addresses the way how human beings develop emotions, based on stimuli from the environment e.g., [14] whereas the second process addresses the way how humans control their emotions in case they do not correspond with the emotions they desire to have e.g., [17].

The current paper focuses on the former, i.e., on emotion generation processes. Its main aim is to present a generic model of emotion generation, which can be used by ambient systems to get insight in the emotion generation processes of a human.

Moreover, the model should be *adaptive*, i.e., it should be able to learn individual characteristics of a person, based on experiences with this person. The idea is that the ambient system observes the *environment* (e.g., which positive and negative events happen?) and the *behaviour* (e.g., which emotional expressions and actions does the human show, and for how long?) of a human in a certain scenario for a certain period, and uses this information to determine the characteristics of this person with respect to emotion generation. Examples of conclusions that the system may draw are “this person is in the process of becoming angry”, or “this person is so angry that (s)he must be calmed down immediately”. This information will allow the system to continuously estimate the emotional state of the human, but also to *predict* its emotional state in future situations. When necessary, it will then use this information for adaptive support. For example, in settings where humans and machines have to cooperate in complex and dynamic environments (e.g., the naval warfare case described above), the system could encourage the human when it predicts he will become sad, or take over some of his tasks when he is becoming angry.

In Section 2, the model for emotion generation is described at a conceptual level, using the modelling language LEADSTO [6]. The idea is that the model is so generic that it can be applied to any arbitrary domain. Section 3 described the model to estimate emotion generation, based on the concept of *Theory of Mind*. In Section 4, a number of simulation results are shown that were generated based on this model. Finally, in Section 5, the model is evaluated and conclusions are drawn.

## 2 A Model for Emotion Generation

In this section, the model for emotion generation will be described at an intuitive, conceptual level, using the agent-based modelling language LEADSTO [6]. This language allows the modeller to integrate both qualitative, logical aspects and quantitative, numerical aspects. In LEADSTO, direct temporal dependencies between two state properties in successive states are modelled by executable dynamic properties. The format is defined as follows: let  $\alpha$  and  $\beta$  be state properties of the form ‘conjunction of ground atoms or negations of ground atoms’. In LEADSTO the notation  $\alpha \rightarrow_{e, f, g, h} \beta$  means:

*If state property  $\alpha$  holds for a certain time interval with duration  $g$ ,  
then after some delay (between  $e$  and  $f$ ) state property  $\beta$  will hold  
for a certain time interval of length  $h$ .*

Here atomic state properties can have a qualitative, logical format, such as an expression `desire(d)`, expressing that desire  $d$  occurs, or a quantitative, numerical format such as an expression `has_value(x, v)` which expresses that variable  $x$  has value  $v$ . For more details, see [6].

In Section 2.1, first a global overview of the model will be provided. Next, Section 2.2 will present the formalisation of the model in LEADSTO.

### 2.1 Emotion Generation Based on a Body Loop

In this and the next section the model to generate emotional states for a given stimulus is introduced. It adopts from [13] the idea of a ‘body loop’ and ‘as if body loop’, but extends this by making these loops recursive. According to the original idea, emotion generation via a body loop roughly proceeds according to the following causal chain; see [7], [13]:

sensing a stimulus  $\rightarrow$  sensory representation of stimulus  $\rightarrow$   
(preparation for) bodily response  $\rightarrow$  sensing the bodily response  $\rightarrow$   
sensory representation of the bodily response  $\rightarrow$  feeling the emotion

As a variation, an ‘as if body loop’ uses a causal relation

preparation for bodily response  $\rightarrow$   
sensory representation of the bodily response

as a shortcut in the causal chain. In the model used here an essential addition is that the body loop (or as if body loop) is extended to a recursive body loop (or recursive as if body loop) by assuming that the preparation of the bodily response is also affected by the state of feeling the emotion (also called emotional feeling):

feeling the emotion  $\rightarrow$  preparation for bodily response

as an additional causal relation. Thus the obtained model is based on reciprocal causation relations between emotional feeling and body states, as roughly shown in Figure 1.




Fig. 1. Recursive body loop

Both the bodily response and the emotional feeling are assigned a level or gradation, expressed by a number, which is assumed dynamic. The causal cycle is modelled as a positive feedback loop, triggered by the stimulus and converging to a certain level of emotional feeling and body state. Here in each round of the cycle the next body state has a level that is affected by both the level of the stimulus and of the emotional feeling state, and the next level of the emotional feeling is based on the level of the body state. In the more detailed model described below, the combined effect of the levels of the stimulus and the emotional state on the body state is modelled as a weighted sum (with equal weights 0.5 in this case). This implies that the pattern of generation (and extinction) of an emotion upon a stimulus is as shown in Figure 2 (where the horizontal axis denotes time and the vertical axis denotes the level of experienced emotion).




Fig. 2. Pattern of emotion generation and extinction by a recursive body loop

## 2.2 Formalisation of the Emotion Generation Model

The specification (both informally and formally) of the model for emotion generation based on a recursive body loop is as follows. This model is based on dynamic Local Properties (LP), expressing the basic mechanisms of the process.

**LP1 Sensing a stimulus**

If a negative stimulus occurs, then a sensor state for the negative stimulus will occur.

$\text{neg\_stimulus} \rightarrow \text{has\_state}(\text{human}, \text{sensor\_state}(\text{neg\_stimulus}))$

**LP2 Generating a sensory representation of a stimulus**

If a sensor state for a negative stimulus occurs,  
then a sensory representation for the negative stimulus will occur.

$\text{has\_state}(\text{human}, \text{sensor\_state}(\text{neg\_stimulus})) \rightarrow \text{has\_state}(\text{human}, \text{srs}(\text{neg\_stimulus}))$

**LP3 From sensory representation and emotion to preparation**

If a sensory representation for a negative stimulus occurs and emotion  $e$  has level  $v$ ,  
then a preparation state for facial expression  $f$  will occur with level  $\alpha * v + (1-\alpha) * 1$

$\text{has\_state}(\text{human}, \text{srs}(\text{neg\_stimulus})) \ \& \ \text{has\_state}(\text{human}, \text{emotion}(e, v)) \rightarrow$   
 $\text{has\_state}(\text{human}, \text{preparation\_state}(f, \alpha * v + (1-\alpha) * 1))$

If no sensory representation for a negative stimulus occurs and emotion  $e$  has level  $v$ ,  
then a preparation state for facial expression  $f$  will occur with level  $v/2$ .

$\text{not has\_state}(\text{human}, \text{srs}(\text{neg\_stimulus})) \ \& \ \text{has\_state}(\text{human}, \text{emotion}(e, v)) \rightarrow$   
 $\text{has\_state}(\text{human}, \text{preparation\_state}(f, v/2))$

**LP4 From preparation to body modification**

If a preparation state for facial expression  $f$  occurs with level  $v$ ,  
then the face is modified to express  $f$  with level  $v$ .

$\text{has\_state}(\text{human}, \text{preparation\_state}(f, v)) \rightarrow \text{has\_state}(\text{human}, \text{effector\_state}(f, v))$

**LP5 From body modification to modified body**

If the face is modified to express  $f$  with level  $v$ ,  
then the face will have expression  $f$  with level  $v$ .

$\text{has\_state}(\text{human}, \text{effector\_state}(f, v)) \rightarrow \text{has\_state}(\text{human}, \text{own\_face}(f, v))$

**LP6 Sensing a body state**

If facial expression  $f$  with level  $v$  occurs,  
then this facial expression is sensed.

$\text{has\_state}(\text{human}, \text{own\_face}(f, v)) \rightarrow \text{has\_state}(\text{human}, \text{sensor\_state}(f, v))$

**LP7 Generating a sensory representation of a body state**

If facial expression  $f$  of level  $v$  is sensed,  
then a sensory representation for facial expression  $f$  with level  $v$  will occur.

$\text{has\_state}(\text{human}, \text{sensor\_state}(f, v)) \rightarrow \text{has\_state}(\text{human}, \text{srs}(f, v))$

**LP8 From sensory representation of body state to emotion**

If a sensory representation for facial expression  $f$  with level  $v$  occurs,  
then emotion  $e$  is felt with level  $v$ .

$\text{has\_state}(\text{human}, \text{srs}(f, v)) \rightarrow \text{has\_state}(\text{human}, \text{emotion}(e, v))$

**3 A Theory of Mind Model to Estimate Emotion Generation**

So far, a model was presented that describes a person's mental states and the relations between them at a global level. However, to be able to provide some intelligent support, an ambient system somehow needs the capability to attribute instances of these mental

states to a human, and to reason about these. In psychology, this capability is often referred to as *Theory of Mind* (or ToM, see, e.g., [4]). According to [9], (human and software) agents can exploit a ToM for two purposes: to *anticipate* the behaviour of other agents (e.g., preparing for certain actions that the other will perform), and to *manipulate* it (e.g., trying to bring the other in a certain state in which he will perform certain desired actions, or not perform certain unwanted actions).

A number of approaches in the literature address the development of formal models for ToM, e.g., [9], [21]. Usually, such models focus on the epistemic (e.g., beliefs) and/or motivational states (e.g., desires, intentions) of other agents. However, since the concept of emotions nowadays is receiving more and more attention, such models ideally also address emotions. This idea is in line with the theories of many Cognitive Scientists like Gärdenfors [15], who claims that humans have a ToM that is not only about beliefs, desires, and intentions, but also about other mental states like emotional and attentional states [15]. Based on these ideas, this paper proposes to apply a ToM to a model for emotion generation as described in the previous section.

### 3.1 Formalisation of a Theory of Mind Model to Estimate Emotion Generation

In order to obtain a Theory of Mind model for generated emotions, the idea of *recursive modelling* is used [21]. This means that the beliefs that agents have about each other are represented in a nested manner. To this end, each mental state is parameterised with the name of the agent that is considered, creating concepts like `has_state(human, emotion(e, 0.5))` and `has_state(AA, performed(remove_neg_stimulus))`. In addition, a number of meta-representations, expressed by meta-predicates are introduced. For example, `has_state(AA, belief(has_state(human, emotion(e, 0.7))))` states that the Ambient Agent (AA) believes that the human has an emotion level of 0.7. The agent AA has also beliefs about relationships between mental states of the human. This is represented in the format:

$$\text{has\_state}(\text{AA}, \text{belief}(\text{leads\_to\_after}(\text{I}, \text{J}, \text{D})))$$

which expresses that when state property I occurs, then after time duration D state property J will occur. An example of a specific case of this is when it is taken:

I	<code>has_state(human, srs(f, v))</code>
J	<code>has_state(human, emotion(e, v))</code>
D	1

With these instances for the variables the representation becomes

$$\text{has\_state}(\text{AA}, \text{belief}(\text{leads\_to\_after}(\text{has\_state}(\text{human}, \text{srs}(\text{f}, \text{v})), \text{has\_state}(\text{human}, \text{emotion}(\text{e}, \text{v})), 1)))$$

This expresses that agent AA believes that when for the human state property `srs(f, v)` occurs, then after one time unit state property `emotion(e, v)` will occur. In such a way it is expressed that agent AA believes that the emotion generation model presented in Section 2 holds. Temporal reasoning based on this model is performed by agent AA using the general reasoning rule

$$\text{has\_state}(\text{AA}, \text{belief}(\text{at}(\text{I}, \text{T}))) \wedge \text{has\_state}(\text{AA}, \text{belief}(\text{leads\_to\_after}(\text{I}, \text{J}, \text{D}))) \rightarrow \text{has\_state}(\text{AA}, \text{belief}(\text{at}(\text{J}, \text{T}+\text{D})))$$

This rule can be considered a temporal forward simulation rule.

### 3.2 Adaptive Aspects in the Model

The model for emotion generation discussed in Section 2 includes a parameter  $\alpha$  for the persistence of an emotional state. The value of such a parameter may not be easy to determine, and may differ between different individuals. Therefore a more realistic approach should include capabilities to adapt and fine tune the value of this parameter. Such a capability has been incorporated in the model. To this end, first the emotion generation model was rewritten to a differential equation model, as follows.

$$\begin{aligned} emotionlevel(t+1) &= \alpha \text{ emotionlevel}(t) + (1 - \alpha) s(t) \\ emotionlevel(t+1) - emotionlevel(t) &= - (1 - \alpha) \text{ emotionlevel}(t) + (1 - \alpha) s(t) \\ &= (1 - \alpha) (s(t) - \text{emotionlevel}(t)) \\ d/dt (s(t) - \text{emotionlevel}(t)) &= - (1 - \alpha) (s(t) - \text{emotionlevel}(t)) \end{aligned}$$

Here  $s(t)$  is the level of the stimulus over time. Note that the time unit here covers exactly one cycle of update of the emotion level (that is 6 time steps in the simulation). This differential equation has the following solution:

$$\begin{aligned} s(t) - \text{emotionlevel}(t) &= (s(0) - \text{emotionlevel}(0)) e^{-(1-\alpha)t} \\ \text{emotionlevel}(t) &= s(t) - (s(0) - \text{emotionlevel}(0)) e^{-(1-\alpha)t} \end{aligned}$$

The next step is to identify the sensitivity of the emotion level with respect to a change of  $\alpha$ . To this end the partial derivative with respect to  $\alpha$  is:

$$\begin{aligned} \partial/\partial\alpha \text{ emotionlevel}(t) &= \partial/\partial\alpha (s(t) - (s(0) - \text{emotionlevel}(0)) e^{-(1-\alpha)t}) \\ &= - (s(0) - \text{emotionlevel}(0)) \partial/\partial\alpha (e^{-(1-\alpha)t}) \\ &= - (s(0) - \text{emotionlevel}(0)) \cdot t \cdot e^{-(1-\alpha)t} \end{aligned}$$

When at time point  $t$  a difference  $d(t) = \Delta \text{ emotionlevel}(t)$  in observed emotion level<sup>1</sup> and calculated emotion level is detected, then based on the derivative w.r.t  $\alpha$  this  $\Delta \text{ emotionlevel}(t)$  can be related to a difference  $\Delta\alpha$  in  $\alpha$ , as follows:

$$\begin{aligned} d(t) = \Delta \text{ emotionlevel}(t) &= \partial/\partial\alpha \text{ emotionlevel}(t) \cdot \Delta\alpha \\ &= - (s(0) - \text{emotionlevel}(0)) \cdot t \cdot e^{-(1-\alpha)t} \cdot \Delta\alpha \end{aligned}$$

So

$$\Delta\alpha = - d(t) / (s(0) - \text{emotionlevel}(0)) \cdot t \cdot e^{-(1-\alpha)t}.$$

This is used in the adaptation process of  $\alpha$  with adaptation speed factor  $\gamma$  as follows:

$$\text{new } \alpha = \alpha + \gamma \Delta\alpha = \alpha - \gamma \cdot d(t) / (s(0) - \text{emotionlevel}(0)) \cdot t \cdot e^{-(1-\alpha)t}$$

When at time 0 the stimulus is 1 and the emotion is 0, this becomes:

$$\text{new } \alpha = \alpha + \gamma \Delta\alpha = \alpha - \gamma \cdot d(t) / t \cdot e^{-(1-\alpha)t}$$

Here, for example,  $\gamma$  can be taken 0.9. If  $t$  and  $d(t)$  are given, the new  $\alpha$  can be calculated using this formula. Within LEADSTO, this mechanism is modelled via the following rules:

#### LP9 Adapt estimated alpha

Adapt the estimation of  $\alpha$  based on the difference in observed and calculated emotion level.

```
real_emotion_available & estimated_alpha(a) & current_time(t) & has_state(human, emotion(e, v1)) & has_state(AA, belief(has_state(human, emotion(e, v)))) →
estimated_alpha(a - gamma * (v1-v2) / (t * 2.71828 ^ (-1 * (1-a) * t)))
```

<sup>1</sup> Hence, this approach assumes that every now and then the real emotion level can be observed.

**LP10 Take over observed emotion level**

If the real emotion level can be observed, use that level in the model (instead of the calculated level).

```
real_emotion_available & has_state(human, emotion(e, v)) →
has_state(AA, belief(has_state(human, emotion(e, v))))
```

**4 Simulation Results**

To test the behaviour of the model to estimate emotion generation, it has been used to perform a number of simulation runs within the LEADSTO simulation environment [6]. The model was tested in a small scenario, involving an ambient agent and a human (indicated by AA and human, respectively). The agent AA was equipped with the model to estimate emotion generation. The central emotion used in the scenario is anger. In order to simulate this, every now and then certain events take place, which influence the level of anger of the human either positively (e.g., a request for an annoying task) or negatively (e.g., the removal of an annoying task from the todo list). The main goal of the agent is to estimate the level of anger of the human. To this end, it starts with some default model of the human's emotion generation dynamics, and then keeps on updating this using the strategies explained earlier. When the human becomes too angry, the ambient agent can take measures to calm him down (e.g., removing an annoying task from the todo list, or taking away an annoying stimulus). In Section 3.1 it was explained how in the model representations of the form

```
has_state(AA, belief(leads_to_after(I, J, D)))
```

together with one forward simulation rule were used. For the sake of simplicity within the simulation the general rule was replaced by instantiated versions, some of which are shown below (i.e., LP12-LP19; note that the explicit temporal dependencies have been left out as well). The first property specifies how the agent AA observes that the human senses a stimulus.

**LP11 Observing human's sensing negative stimulus**

If the human senses a negative stimulus then the ambient agent AA will observe this.

```
has_state(human, sensor_state(neg_stimulus)) →
has_state(AA, observed(has_state(human, sensor_state(neg_stimulus))))
```

**LP12 Belief generation of human's sensing negative stimulus**

If the ambient agent observes that the human senses a negative stimulus, then it will generate a belief on it.

```
has_state(AA, observed(has_state(human, sensor_state(neg_stimulus)))) →
has_state(AA, belief(has_state(human, sensor_state(neg_stimulus))))
```

**LP13 Generating a sensory representation of human sensing negative stimulus**

If AA believes that the human senses a negative stimulus, then it will generate a belief that the human will have a sensory representation for this stimulus.

```
has_state(AA, belief(has_state(human, sensor_state(neg_stimulus)))) →
has_state(AA, belief(has_state(human, srs(neg_stimulus))))
```

**LP14 From sensory representation and emotion to preparation**

If AA believes that the human has a sensory representation for a negative stimulus and AA believes that the human has emotion  $e$  with level  $v$ , then it will generate the belief that the human's preparation state for facial expression  $f$  will occur with level  $\alpha * v + (1-\alpha) * 1$ .

```
has_state(AA, belief(has_state(human, srs(neg_stimulus)))) &
has_state(AA, belief(has_state(human, emotion(e, v)))) →
has_state(AA, belief(has_state(human, preparation_state(f,  $\alpha * v + (1-\alpha) * 1$ ))))
```

If AA believes that the human has NO sensory representation for a negative stimulus and AA believes that the human has emotion  $e$  with level  $v$ , then it will generate the belief that the human's preparation state for facial expression  $f$  will occur with level  $v/2$ .

```
not(has_state(AA, belief(has_state(human, srs(neg_stimulus)))) &
has_state(AA, belief(has_state(human, emotion(e, v)))) →
has_state(AA, belief(has_state(human, preparation_state(f, v/2))))
```

**LP15 From preparation to body modification**

If AA believes that the human's preparation state for facial expression  $f$  with level  $v$  occurred, then it will believe that the humans' face is modified to express  $f$  with level  $v$ .

```
has_state(AA, belief(has_state(human, preparation_state(f, v)))) →
has_state(AA, belief(has_state(human, effector_state(f, v))))
```

**LP16 From body modification to modified body**

If AA believes that the human's face is modified to express  $f$  with level  $v$ , then it will believe that the human's face will have expression  $f$  with level  $v$ .

```
has_state(AA, belief(has_state(human, effector_state(f, v)))) →
has_state(AA, belief(has_state(human, own_face(f, v))))
```

**LP17 Sensing a body state**

If AA believes that the human's face has expression  $f$  with level  $v$ , then it will believe that the human will sense this facial expression.

```
has_state(AA, belief(has_state(human, own_face(f, v)))) →
has_state(AA, belief(has_state(human, sensor_state(f, v))))
```

**LP18 Generating a sensory representation of a body state**

If AA believes that the human has sensed facial expression  $f$  with level  $v$ , then it will believe that the human has a sensory representation for facial expression  $f$  with level  $v$ .

```
has_state(AA, belief(has_state(human, sensor_state(f, v)))) →
has_state(AA, belief(has_state(human, srs(f, v))))
```

**LP19 From sensory representation of body state to emotion**

If AA believes that the human has a sensory representation for facial expression  $f$  with level  $v$ , then it will believe that the human has emotion  $e$  with level  $v$ .

```
has_state(AA, belief(has_state(human, srs(f, v)))) →
has_state(AA, belief(has_state(human, emotion(e, v))))
```

In addition, a number of other rules have been established to model the behaviour of the human and the ambient agent, and its effect on the world:

**LP20 Intervention by the Ambient Agent**

If AA believes that the human has emotion  $e$  with level  $v$  which is higher than a certain threshold  $th1$ , then it will remove a negative stimulus.

$has\_state(AA, belief(has\_state(human, emotion(e, v)))) \ \& \ v \geq th1 \rightarrow has\_state(AA, performed(remove\_neg\_stimulus))$

**LP21 Effect of intervention in the world**

As long as AA does not remove a negative stimulus, it persists in the world.

$not \ has\_state(AA, performed(remove\_neg\_stimulus)) \rightarrow neg\_stimulus$

**LP22 Performance of human**


If the human has emotion  $e$  with level  $v$  which is higher than a certain threshold  $th2$ , then (s)he will show bad performance.

$has\_state(human, emotion(e, v)) \ \& \ v \geq th2 \rightarrow has\_state(human, performed(bad\_performance))$

If the human has emotion  $e$  with level  $v$  which is lower than threshold  $th2$ , then (s)he will show good performance.

$has\_state(human, emotion(e, v)) \ \& \ v < th2 \rightarrow has\_state(human, performed(good\_performance))$

Based on the model, a number of simulations (under different parameter settings) have been performed, and some of the simulation traces are included in this section for analysis; see Figure 3 to Figure 6. In all of these figures, where time is on the horizontal axis, the upper part shows the time periods, in which the binary logical



**Fig. 3.** Simulation Trace 1 - Estimated  $\alpha$  is equal to real  $\alpha$




Fig. 4. Simulation Trace 2 - Estimated  $\alpha$  is higher than real  $\alpha$

state properties hold (indicated by the dark lines); for example; `neg_stimulus`, `estimated_alpha(X)`, and `has_state(AA, belief(has_state(human, sensor_state(neg_stimulus))))`. Below this part, quantitative information is provided about the human's actual emotion level, and the Ambient Agent's estimation of this emotion level, respectively. Values for these levels for the different time periods are shown by the dark lines. For example, in Figure 3, at time point 10 AA estimates that the human's emotion level is 0.5, but this increases to 0.75 at time point 15 and further. The graphs show how the recursive body loop approximates a state for emotion with value 1. Note that only a selection of relevant state properties is shown.

*Trace 1*

This trace (see Figure 3), shows a normal situation, in which the estimated  $\alpha$  is equal to the real  $\alpha$  indicated in the upper part of the Figure 3, by state properties `estimated_alpha(0.5)` and `alpha(0.5)` respectively. As shown in the figure, the Ambient Agent removes the negative stimulus exactly at the right moments (i.e., at time point 22, 59, and 95). As a result, the human never shows bad performance. Note that, in this trace (as well as the next two), threshold `th1` for intervention (see LP15) was set to 0.8, and threshold `th2` for negative performance (see LP17) was set to 0.95.




Fig. 5. Simulation Trace 3 - Estimated  $\alpha$  is lower than real  $\alpha$

*Trace 2*

This trace (see Figure 4), shows a situation in which the estimated  $\alpha$  (0.8) is higher than the real  $\alpha$  (0.5), as indicated in the upper part of the Figure 4. As shown in the figure, the Ambient Agent estimates the level of emotion of the human much too low, so that it is too late in removing the negative stimulus, indicated in the upper part by state property `has_state(AA, performed(remove_neg_stimulus))` at time point 52. This is too late, because, as shown in the “actual emotion” graph below, the human’s emotion level has gone too high already at time point 32. As a result, the human shows bad performance at time point 33 (as indicated by the state property `has_state(human, performed(bad_performance))` in the upper part of Figure 4).

*Trace 3*

This trace (see Figure 5) shows a situation in which the estimated  $\alpha$  (0.2) is lower than the real  $\alpha$  (0.5), as indicated in the upper part of the Figure 5. As shown in the figure, the Ambient Agent estimates the level of emotion of the human too high, so that it is a bit too early in removing the negative stimulus, as indicated in the upper part by property `has_state(AA, performed(remove_neg_stimulus))` at time point 11. This is not a crucial error (the human does not show bad performance), but it is a waste of energy.

*Trace 4*

This trace (see Figure 6) shows a situation in which the estimated  $\alpha$  is learned. In this trace, speed factor  $\gamma$  (see LP9) was set to 0.9, threshold `th1` was set to 0.5, and `th2` was




Fig. 6. Simulation Trace 4 - Estimated  $\alpha$  is adapted

set to 0.95. As shown in the figure, the Ambient Agent initially estimates the level of emotion of the human much too low, indicated in the upper part by the state property `estimated_alpha(0.8)`. As a result, at time point 32, the human’s emotion level has gone too high (shown in the “actual emotion” graph below), so that AA is too late in removing the negative stimulus. However, after information about the real emotion became available (at time point 35) (shown by state property `real_emotion_available`), AA changes its estimation of  $\alpha$  (from 0.8 to 0.65). As a result, from then on the agent estimates the emotion much better, and removes the stimulus at the right moments (at time points 67 and 93), so that the human does not show bad performance anymore.

All in all, the model has been used to generate a large number of simulation traces, using different parameter settings for the real and estimated parameter  $\alpha$ . Due to space limitations, not all of these results are shown here. However, the simulation experiments pointed out that the model in general is successful in estimating a (simulated) person’s emotion generation dynamics, and is robust to different parameter settings.

The obvious next step is to test the model in a real world setting. This can be done, for example, in a laboratory experiment where a person is asked to perform a computer task during which certain emotion-eliciting events occur. To measure the person’s emotional reactions to these events, recent methods to recognise levels of

emotion based on facial expressions may be employed e.g., [11], [12], [16], [18], [22], [24].

## 5 Discussion

To improve the performance and wellbeing of humans in complex human-computer interaction settings, ambient systems need to assess many aspects of the human's state. In addition to its state of awareness, stress, and motivation, the system needs to assess the human's emotional state, and more specifically, its emotion generation capabilities. One step further, the system requires the ability to reason about these emotion generation capabilities, and to make predictions based on them. For example, if the system knows that the human is very quick in developing a state of anger, this will be useful to determine how to communicate with the user for the next hours.

As a first step in this direction, the current paper introduces an adaptive computational model to estimate emotion generation processes. The model combines two main components, namely a model for emotion generation (inspired by [13]) and a model for Theory of Mind cf. [21]. The model has been implemented using the modelling language LEADSTO, and has been tested under different parameter settings.

The model is based on several important assumptions. For one, it is assumed that it is possible to measure a person's emotional reactions to certain events. The simulation runs, abstract from a specific technique that may be used for this, but in real world experiments, obviously this cannot be done anymore. For future research the plan is to test the model in laboratory experiments; here emotion recognition approaches like [11], [12], [16], [18], [22], [24] may be used. In recent years, such approaches have proven to be very adequate in recognising emotion *elicitation* processes in humans. Future work should point out how different types of approaches can be compared. One of these differences may be the possibility to reason over time about the emotion generation process, for example, to predict future emotions, or to predict effects of a certain intervention on the emotion level.

Another interesting direction for further research is to explore the possibilities to apply the model to computer-computer interaction instead of human-computer interaction settings. For example, a recent trend in the development of *intelligent virtual agents* (IVAs, see [23]) is to equip them with emotions and emotion generation mechanisms (e.g., [5], [20]). As soon as such IVAs start to communicate with each other, it will be useful for them to have insight in each other's emotions, and to make predictions about them. It may be expected that the generic setup of the model presented here allows it to be equally well applicable to software agents as to humans. Therefore, it may eventually be applied in larger systems where a number of real humans and virtual humans have to cooperate.

## References

1. Aarts, E., Collier, R., van Loenen, E., de Ruyter, B.: EUSAI 2003. LNCS, vol. 2875, p. 432. Springer, Heidelberg (2003)
2. Aarts, E., Harwig, R., Schuurmans, M.: Ambient Intelligence. In: Denning, P. (ed.) *The Invisible Future*, pp. 235–250. McGraw-Hill, New York (2001)

3. Ball, G., Breese, J.: Modelling the emotional state of computer users. In: Proceedings of the Workshop on Personality and Emotion in User Modelling (1999)
4. Baron-Cohen, S.: Mindblindness. MIT Press, Cambridge (1995)
5. Bates, J.: The role of emotion in believable agents. *Communications of the ACM* 37(7), 122–125 (1994)
6. Bosse, T., Jonker, C.M., van der Meij, L., Treur, J.: A Language and Environment for Analysis of Dynamics by Simulation. *International Journal of Artificial Intelligence Tools* 16(3), 435–464 (2007)
7. Bosse, T., Jonker, C.M., Treur, J.: Formalisation of Damasio's Theory of Emotion, Feeling and Core Consciousness. *Consciousness and Cognition Journal* (in press, 2008); Fum, D., Del Missier, F., Stocco, A. (eds.), Proc. of the 7th International Conference on Cognitive Modelling, ICCM 2006, pp. 68–73 (2006)
8. Bosse, T., van Maanen, P.P., Treur, J.: A Cognitive Model for Visual Attention and its Application. In: Nishida, T., Klusch, M., Sycara, K., Yokoo, M., Liu, J., Wah, B., Cheung, W., Cheung, Y.-M. (eds.) Proceedings of the Sixth International Conference on Intelligent Agent Technology, IAT 2006, pp. 255–262. IEEE Computer Society Press, Los Alamitos (2006)
9. Bosse, T., Memon, Z.A., Treur, J.: A Two-Level BDI-Agent Model for Theory of Mind and its Use in Social Manipulation. In: Proceedings of the AISB 2007 Workshop on Mindful Environments, pp. 335–342 (2007)
10. Bosse, T., Pontier, M., Treur, J.: A Dynamical System Modelling Approach to Gross' Model of Emotion Regulation. In: Lewis, R.L., Polk, T.A., Laird, J.E. (eds.) Proceedings of the Eighth International Conference on Cognitive Modeling, ICCM 2007, pp. 187–192. Taylor and Francis, Abington (2007)
11. Busso, C., Deng, Z., Yildirim, S., Bulut, M., Lee, C.M., Kazemzadeh, A., Lee, S., Neumann, U., Narayanan, S.: Analysis of emotion recognition using facial expressions, speech and multimodal information. In: Proceedings of the ACM Sixth International Conference on Multimodal Interfaces, ICMI 2004, State College, PA (2004)
12. Cohen, I., Garg, A., Huang, T.S.: Emotion recognition using multilevel HMM. In: Proceedings of the NIPS Workshop on Affective Computing, Colorado (2000)
13. Damasio, A.: *The Feeling of What Happens: Body, Emotion and the Making of Consciousness*. Harcourt Brace (1999)
14. Frijda, N.H., Manstead, A.S.R., Bem, S. (eds.): *Emotions and Beliefs: How Feelings Influence Thoughts*. Cambridge Univ. Press, NY (2000)
15. Gärdenfors, P.: Slicing the Theory of Mind. In: Danish yearbook for philosophy, vol. 36, pp. 7–34. Museum Tusulanum Press (2001)
16. Goldman, A.I., Sripada, C.S.: Simulationist models of face-based emotion recognition. *Cognition* 94, 193–213 (2005)
17. Gross, J.J.: The Emerging Field of Emotion Regulation: An Integrative Review. *Review of General Psychology* 2(3), 271–299 (1998)
18. Ioannou, S.V., Raouzaiou, A.T., Tzouvaras, V.A., Mailis, T.P., Karpouzis, K.C., Kollias, S.D.: Emotion Recognition Through Facial Expression Analysis Based on a Neurofuzzy Network. *Neural Networks* 18, 423–435 (2005)
19. Kaber, D.B., Endsley, M.R.: The effects of level of automation and adaptive automation on human performance, situation awareness and workload in a dynamic control task. *Theoretical Issues in Ergonomics Science* 5(2), 113–153 (2004)
20. Marsella, S., Gratch, J.: Modeling coping behavior in virtual humans: Don't worry, be happy. In: Proceedings of Second International Joint Conference on Autonomous Agents and Multiagent Systems, AAMAS 2003, pp. 313–320. ACM Press, New York (2003)

21. Marsella, S.C., Pynadath, D.V., Read, S.J.: PsychSim: Agent-based modeling of social interaction and influence. In: Lovett, M., et al. (eds.) Proceedings of ICCM 2004, Pittsburg, Pennsylvania, USA, pp. 243–248 (2004)
22. Pantic, M., Rothkrantz, L.J.M.: Automatic Recognition of Facial Expressions and Human Emotions. In: Proceedings of ASCI 1997 conference, ASCI, Delft, pp. 196–202 (1997)
23. Pelachaud, C., Martin, J.C., Andre, E., Chollet, G., Karpouzis, K., Pele, D.: IVA 2007. LNCS (LNAI), vol. 4722. Springer, Heidelberg (2007)
24. Vogt, T., André, E.: Comparing Feature Sets for Acted and Spontaneous Speech in View of Automatic Emotion Recognition. In: Proceedings of the IEEE International Conference on Multimedia & Expo, ICME 2005, Amsterdam, the Netherlands (2005)